


Chapter 4

metalinguistic Abstraction


What if Scheme is not enough?



Metalinguistic abstraction -- establishing new languages -- plays an important role in all branches of engineering design. It is particularly important to computer programming, because in programming not only can we formulate new languages but we can also implement these languages by constructing evaluators. An evaluator (or interpreter) for a programming language is a procedure that, when applied to an expression of the language, performs the actions required to evaluate that expression.

The interpreter is just another program


The Interpreter is just another program



Keep the environment model in mind

Concrete vs. Abstract Syntax

```
static int fact(int n) {  
    if (n <= 1) {  
        return 1;  
    }  
    else {  
        return n * fact(n-1);  
    }  
}
```



Concrete vs. Abstract Syntax

Same example in Scheme

Turn program into
data structure

```
(define (fact n)
  (if (<= n 1)
      1
      (* n (fact (- n 1)))))
```



```
'(define (fact n)
  (if (<= n 1)
      1
      (* n (fact (- n 1)))))
```

The Scheme evaluator is just
a **list processing program!**

The Scheme Evaluator: eval

special forms
are special

```
(define (eval exp env)
  (cond ((self-evaluating? exp) exp)
        ((variable? exp) (lookup-variable-value exp env))
        ((quoted? exp) (text-of-quotation exp))
        ((assignment? exp) (eval-assignment exp env))
        ((definition? exp) (eval-definition exp env))
        ((if? exp) (eval-if exp env))
        ((lambda? exp)
         (make-procedure (lambda-parameters exp)
                         (lambda-body exp)
                         env))
        ((begin? exp)
         (eval-sequence (begin-actions exp) env))
        ((cond? exp) (eval (cond->if exp) env))
        ((application? exp)
         (apply (eval (operator exp) env)
                (list-of-values (operands exp) env))))
        (else
         (error "Unknown expression type -- EVAL" exp))))
```

call more specific
evaluation procedures

syntactic
sugar

apply is the
regular case

The Scheme Evaluator: apply

already
evaluated!

```
(define (apply procedure arguments)
  (cond ((primitive-procedure? procedure)
         (apply-primitive-procedure procedure arguments))
        ((compound-procedure? procedure)
         (eval-sequence
          (procedure-body procedure) ← body
          (extend-environment
           (procedure-parameters procedure)
           arguments
           (procedure-environment procedure))))))
  (else
   (error
    "Unknown procedure type -- APPLY" procedure))))
```

local environment

```
(define (list-of-values exps env)
  (if (no-operands? exps)
      '()
      (cons (eval (first-operand exps) env)
            (list-of-values (rest-operands exps) env)))))
```

Representing Expressions (I)

```
(define (self-evaluating? exp)
  (cond ((number? exp) #t)
        ((string? exp) #t)
        (else #f)))
```

```
(define (variable? exp) (symbol? exp))
```

Simple expressions

```
(define (tagged-list? exp tag)
  (if (pair? exp)
      (eq? (car exp) tag)
      #f))
```

Compound
expressions

First
example
(quote exp)

```
(define (quoted? exp)
  (tagged-list? exp 'quote))
```

Accessor

```
(define (text-of-quotation exp) (cadr exp))
```

Representing Expressions (2)

```
(define (assignment? exp)
  (tagged-list? exp 'set!))
(define (assignment-variable exp) (cadr exp))
(define (assignment-value exp) (caddr exp))
```

(set! var val)

```
(define (definition? exp)
  (tagged-list? exp 'define))
(define (definition-variable exp)
  (if (symbol? (cadr exp))
      (cadr exp)
      (caadr exp)))
(define (definition-value exp)
  (if (symbol? (cadr exp))
      (caddr exp)
      (make-lambda (cdadr exp) ; formal parameters
                  (cddr exp)))) ; body
(define (make-lambda parameters body)
  (cons 'lambda (cons parameters body)))
```

(define var val)

(define (var ...) val)

Syntactic sugar

Representing Expressions (3)

```
(define (make-procedure parameters body env)
  (list 'procedure parameters body env))
```

```
(define (compound-procedure? p)
  (tagged-list? p 'procedure))
```

```
(define (procedure-parameters p) (cadr p))
(define (procedure-body p) (caddr p))
(define (procedure-environment p) (cadddr p))
```

Representing Expressions (4)

```
(define (lambda? exp) (tagged-list? exp 'lambda))  
(define (lambda-parameters exp) (cadr exp))  
(define (lambda-body exp) (caddr exp))
```

(lambda pars body)

```
(define (if? exp) (tagged-list? exp 'if))  
(define (if-predicate exp) (cadr exp))  
(define (if-consequent exp) (caddr exp))  
(define (if-alternative exp)  
  (if (not (null? (cdddr exp)))  
      (cadddr exp)  
      'false))
```

(if pred cons alt)

```
(define (make-if predicate consequent alternative)  
  (list 'if predicate consequent alternative))
```

Evaluating cond
syntactic sugar

Representing Expressions (5)

```
(define (begin? exp) (tagged-list? exp 'begin))
(define (begin-actions exp) (cdr exp))
(define (last-exp? seq) (null? (cdr seq)))
(define (first-exp seq) (car seq))
(define (rest-exps seq) (cdr seq))
```

Evaluating
syntactic sugar

```
(define (sequence->exp seq)
  (cond ((null? seq) seq)
        ((last-exp? seq) (first-exp seq))
        (else (make-begin seq))))
(define (make-begin seq) (cons 'begin seq))
```

```
(define (application? exp) (pair? exp))
(define (operator exp) (car exp))
(define (operands exp) (cdr exp))
(define (no-operands? ops) (null? ops))
(define (first-operand ops) (car ops))
(define (rest-operands ops) (cdr ops))
```

The Evaluator Parts

```
(define (eval-assignment exp env)
  (set-variable-value! (assignment-variable exp)
    (eval (assignment-value exp) env)
    env)
  'ok)

(define (eval-definition exp env)
  (define-variable! (definition-variable exp)
    (eval (definition-value exp) env)
    env)
  'ok)

(define (eval-if exp env)
  (if (true? (eval (if-predicate exp) env))
    (eval (if-consequent exp) env)
    (eval (if-alternative exp) env)))

(define (eval-sequence exps env)
  (cond ((last-exp? exps) (eval (first-exp exps) env))
        (else (eval (first-exp exps) env)
              (eval-sequence (rest-exp exps) env)))))
```

Remember...

```
(define (eval exp env)
  (cond ((self-evaluating? exp) exp)
        ((variable? exp) (lookup-variable-value exp env))
        ((quoted? exp) (text-of-quotation exp))
        ((assignment? exp) (eval-assignment exp env))
        ((definition? exp) (eval-definition exp env))
        ((if? exp) (eval-if exp env))
        ((lambda? exp)
         (make-procedure (lambda-parameters exp)
                        (lambda-body exp)
                        env))
        ((begin? exp)
         (eval-sequence (begin-actions exp) env))
        ((cond? exp) (eval (cond->if exp) env))
        ((application? exp)
         (apply (eval (operator exp) env)
                (list-of-values (operands exp) env))))
        (else
         (error "Unknown expression type -- EVAL" exp))))
```

Syntactic sugar

syntactic sugar

```
(define (cond? exp) (tagged-list? exp 'cond))
(define (cond-clauses exp) (cdr exp))
(define (cond-else-clause? clause)
  (eq? (cond-predicate clause) 'else))
(define (cond-predicate clause) (car clause))
(define (cond-actions clause) (cdr clause))

(define (cond->if exp)
  (expand-clauses (cond-clauses exp)))
(define (expand-clauses clauses)
  (if (null? clauses)
      'false
      ; no else clause
      (let ((first (car clauses))
            (rest (cdr clauses)))
        (if (cond-else-clause? first)
            (if (null? rest)
                (sequence->exp (cond-actions first))
                (error "ELSE clause isn't last -- COND->IF"
                      clauses))
            (make-if (cond-predicate first)
                  (sequence->exp (cond-actions first))
                  (expand-clauses rest))))))
```

Environments are lists of frames

Each frame of an environment is represented as a pair of lists: a list of the variables bound in that frame and a list of the associated values

```
(define (make-frame variables values)
  (cons variables values))
(define (frame-variables frame) (car frame))
(define (frame-values frame) (cdr frame))
(define (add-binding-to-frame! var val frame)
  (set-car! frame (cons var (car frame)))
  (set-cdr! frame (cons val (cdr frame))))
```

Variables with their corresponding values

```
(define the-empty-environment '())
(define (enclosing-environment env) (cdr env))
(define (first-frame env) (car env))
```

Lists of Frames

Environments are lists of frames

To extend an environment by a new frame that associates variables with values, we make a frame consisting of the list of variables and the list of values, and we adjoin this to the environment. We signal an error if the number of variables does not match the number of values.

```
(define (extend-environment vars vals base-env)
  (if (= (length vars) (length vals))
      (cons (make-frame vars vals) base-env)
      (if (< (length vars) (length vals))
          (error "Too many arguments supplied" vars vals)
          (error "Too few arguments supplied" vars vals))))
```

lookup Identifiers

To look up a variable in an environment, we scan the list of variables in the first frame. If we find the desired variable, we return the corresponding element in the list of values. If we do not find the variable in the current frame, we search the enclosing environment, and so on. If we reach the empty environment, we signal an “unbound variable” error.

```
(define (lookup-variable-value var env)
  (define (env-loop env)
    (define (scan vars vals)
      (cond ((null? vars)
              (env-loop (enclosing-environment env)))
            ((eq? var (car vars))
              (car vals))
            (else (scan (cdr vars) (cdr vals)))))
        (if (eq? env the-empty-environment)
            (error "Unbound variable" var)
            (let ((frame (first-frame env)))
              (scan (frame-variables frame)
                    (frame-values frame)))))

  (env-loop env))
```

lookup in
current frame

lookup in
next frame

Mutual recursive
search process

Ended by a read

Overwriting Identifiers

To set a variable to a new value in a specified environment, we scan for the variable, just as in `lookup-variable-value`, and change the corresponding value when we find it.

```
(define (set-variable-value! var val env)
  (define (env-loop env)
    (define (scan vars vals)
      (cond ((null? vars)
              (env-loop (enclosing-environment env)))
            ((eq? var (car vars))
              (set-car! vals val))
            (else (scan (cdr vars) (cdr vals))))))
    (if (eq? env the-empty-environment)
        (error "Unbound variable -- SET!" var)
        (let ((frame (first-frame env)))
          (scan (frame-variables frame)
                (frame-values frame)))))
  (env-loop env))
```

lookup in current frame

lookup in next frame

Mutual recursive search process

Ended by a write

Adding Identifiers

To define a variable, we search the first frame for a binding for the variable, and change the binding if it exists (just as in `set-variable!`). If no such binding exists, we adjoin one to the first frame.

```
(define (define-variable! var val env)
  (let ((frame (first-frame env)))
    (define (scan vars vals)
      (cond ((null? vars)
              (add-binding-to-frame! var val frame))
            ((eq? var (car vars))
              (set-car! vals val))
            (else (scan (cdr vars) (cdr vals))))))
    (scan (frame-variables frame)
          (frame-values frame))))
```

lookup in
current frame

Remember

```
(define (apply procedure arguments)
  (cond ((primitive-procedure? procedure)
         (apply-primitive-procedure procedure arguments))
        ((compound-procedure? procedure)
         (eval-sequence
          (procedure-body procedure)
          (extend-environment
           (procedure-parameters procedure)
           arguments
           (procedure-environment procedure))))))
  (else
   (error
    "Unknown procedure type -- APPLY" procedure)))))

(define (list-of-values exps env)
  (if (no-operands? exps)
      '()
      (cons (eval (first-operand exps) env)
            (list-of-values (rest-operands exps) env)))))
```

Primitive Procedures

```
(define (primitive-procedure? proc)
  (tagged-list? proc 'primitive))
(define (primitive-implementation proc) (cadr proc))
(define primitive-procedures
  (list (list 'car car)
        (list 'cdr cdr)
        (list 'cons cons)
        (list 'null? null?)))
(define (primitive-procedure-names)
  (map car
       primitive-procedures))
(define (primitive-procedure-objects)
  (map (lambda (proc) (list 'primitive (cadr proc)))
       primitive-procedures))


(define (apply-primitive-procedure proc args)
  (apply-in-underlying-scheme
    (primitive-implementation proc) args))
```

Built-in apply

Some IO Tools

```
(define (driver-loop)
  (prompt-for-input input-prompt)
  (let ((input (read)))
    (let ((output (eval input the-global-environment)))
      (announce-output output-prompt)
      (user-print output)))
  (driver-loop))

(driver-loop)
```



Go into an
infinite loop

Some IO Tools

```
(define input-prompt ";; M-Eval input:")
(define output-prompt ";; M-Eval value:")

(define (prompt-for-input string)
  (newline) (newline) (display string) (newline))

(define (announce-output string)
  (newline) (display string) (newline))

(define (user-print object)
  (if (compound-procedure? object)
      (display (list 'compound-procedure
                     (procedure-parameters object)
                     (procedure-body object)
                     '<procedure-env>))
      (display object)))
```



Don't show the
environment

let us Test our Machine

The screenshot shows the DrRacket IDE interface. The title bar reads "m-eval.rkt – DrRacket". The menu bar has "m-eval.rkt" and "(define ...)" dropdowns. The toolbar includes "Check Syntax" (with a red error icon), "Debug" (with a green play icon), "Macro Stepper" (with a blue step icon), "Run" (with a green play icon), and "Stop" (with a red square icon). The code editor contains the following Racket code:

```
(procedure-parameters object)
(procedure-body object)
'<procedure-env>))
(display object)))
```

The welcome message in the main window says:

Welcome to [DrRacket](#), version 5.3.1 [3m].
Language: [R5RS \[custom\]](#); memory limit: 128 MB.

The M-Eval input field contains ";;; M-Eval input:" followed by an empty text area with a yellow "eof" button on the right. The status bar at the bottom left says "R5RS custom" and the bottom right shows "6:2" and a small logo.

The Global Environment

```
(define (setup-environment)
  (let ((initial-env
         (extend-environment (primitive-procedure-names)
                             (primitive-procedure-objects)
                             the-empty-environment)))
    (define-variable! 'true #t initial-env)
    (define-variable! 'false #f initial-env)
    initial-env))

(define the-global-environment (setup-environment))
```

Wrap Up Chapter 4

- The interpreter is just another program
 - concrete vs. abstract syntax
 - the interpreter is a list processor
- Meta-circular evaluators
 - recursive interplay eval/apply
 - syntactic sugar = list transformation
 - environments are lists of frames